



作者：蔡宗成、陳明周(2001-06-20)；推薦：徐業良(2001-06-21)。

## 單晶片實習－VB 與 RS232 之應用

本文是介紹如何使用 VB 程式撰寫簡易 RS232 溝程式，傳輸控制指令到 89C51 單晶片上，輸出數位訊號，控制周邊設備，例如 LED 閃爍、步進馬達正反轉等。

### 1. 89C51 單晶片基本電路

圖 1 所示為 89C51 單晶片基本電路設計，供應單晶片 +5V 工作電壓，接上 12MHz 石英震盪器作為時脈電路設計，RESET 按鈕提供系統重置電路，PC 之 RS232 必須經過 MAX232（或其相容單晶片）電位轉換單晶片，才能與 89C51 之串列埠連接，確保傳輸資料正確。




圖 1. 單晶片基本電路

圖 2 所示為 89C51 單晶片輸出電路，使用 PORT0 之 P0.0~P0.4(pin39~pin35)，相關腳位功能與說明如表 1 所示。




圖 2. 輸出電路

表 1. 腳位說明

腳位	電路	功能
P0.0	LED與220歐姆	設定RS232顯示燈
P0.1	LED與220歐姆	接收資料顯示燈
P0.2	LED與220歐姆	明、暗
P0.3	LED與220歐姆	每0.5秒閃爍一次
P0.4	LED與220歐姆	每1秒閃爍一次

圖 2 所示為基本輸出電路圖，PORT0 先接 LED 燈再接 220 歐姆到 VCC 供應電壓，接電阻是防止 LED 燈燒毀，控制 PORT0 之接腳輸出為 0（低電位）時，VCC 電壓流經 LED 燈，使之導通點亮，反之，當 PORT0 之接腳輸出為 1（高電位）時，LED 燈便會斷路而滅。本文先介紹基本輸出電路，以 VB 撰寫 RS232 溝通程式，輸出控制指令，使 LED 燈產生明、暗、閃爍等動作，未來可以此為基本架構，發展更多的功能，例如加入放大電路來控制步進馬達，或是驅動繼電器來控制電器開關，更可以加入感測裝置，感測溫度、亮度等，再將資訊傳回電腦，做進一步的應用與處理。

## 2. 89C51 組合語言

圖 3 所示為 89C51 單晶片之控制程式，總共分成三個部份，第一部份為程式主體，如圖 3(a)所示，程式一開始執行，便設定 RS232 之相關通訊協定，以 P0.0 之 LED 燈的明、暗表示設定中與設定完畢，接著，開始等待 RS232 有無傳送資料過來，以 P0.1 之 LED 燈的明、暗表示接收中與接收完畢。圖 3(b)為資料判斷程式，當程式接收傳來的資料時，開始判斷資料形式，逐一比對，接收到正確的指令時，使 LED 燈產生所需要的動作。圖 3(c)為副程式，分別為設定 RS232 副程式與時間延遲程式，此副程式設計有助於縮短程式開發時間，使一個較大的程式分割成幾個較小的副程式，幫助設計者容易開發程式、減少除錯時間，另一個好處是可直接把這些副程式直接套用在另一個程式上。

```

;RS232d---select case-and-RS232---2001/06/18---蔡宗成;
;*****;
START:  ORG    00H          ;初始設定
        JMP    MAIN
        ORG    30H
MAIN:   MOV    SP,#5FH
;-----;
        CALL   INIT_RS232 ;呼叫INIT_RS232副程式，設定串列埠的通訊協定
        CPL    P0.0        ;P0.0亮，表示正在設定串列埠（CPL指令表示反轉腳位狀態）
        MOV    R5,#25      ;設定R5=25，延遲25*20ns = 500ns = 0.5秒
        CALL   DELAYms     ;呼叫DELAYms副程式
        CPL    P0.0        ;P0.0暗，表示設定完成
;-----;
WAIT:   JNB    RI,WAIT     ;等待程式，檢查RI是否為1？RI=1表示收到資料，往下執行
        CPL    P0.1        ;P0.1亮，表示正在接收，開始比對
        CLR    RI          ;清除RI以便繼續接收。
        MOV    A,SBUF      ;從串列埠讀入資料
;-----;

```

圖 3(a). 主程式

```

;--SELECT CASE-----;
;--CASE "A"-----;以下比對 A(SBUF) 的資料形式
MA:
        CJNE   A,#1,MB    ;CASE-A，收到"1"往下執行；否則跳到MB
        CPL    P0.2        ;P0.2 亮
        JMP    MEND       ;執行結束，跳到MEND
;--CASE "B"-----;
MB:
        CJNE   A,#2,MC    ;CASE-B，收到"2"往下執行；否則跳到MC
        MOV    R1,#10     ;設定R1=10，執行閃爍程式10次
MBLOOP:
        CPL    P0.3        ;P0.3 每0.5秒閃爍，共10次
        MOV    R5,#25     ;設定R5=25，延遲25*20ns = 500ns = 0.5秒
        CALL   DELAYms    ;呼叫DELAYms副程式
        CPL    P0.3        ;
        MOV    R5,#25     ;設定R5=25，延遲25*20ns = 500ns = 0.5秒
        CALL   DELAYms    ;
        DJNZ   R1,MBLOOP  ;R1-1，判斷R1是否為0？跳回MBLOOP，執行回圈10次
        JMP    MEND       ;執行結束，跳到MEND
;--CASE "C"-----;
MC:
        CJNE   A,#3,MEND  ;CASE-C，收到"3"往下執行；否則跳到MEND
        MOV    R1,#10     ;設定R1=10，執行閃爍程式10次
MCLOOP:
        CPL    P0.4        ;P0.4亮，每1秒閃爍，共10次
        MOV    R5,#50     ;設定R5=50，延遲50*20ns = 1000ns = 1秒
        CALL   DELAYms    ;呼叫DELAYms副程式
        CPL    P0.4        ;
        MOV    R5,#50     ;設定R5=50，延遲50*20ns = 1000ns = 1秒
        CALL   DELAYms    ;
        DJNZ   R1,MCLOOP  ;R1-1，判斷R1是否為0？跳回MCLOOP，執行回圈10次
        JMP    MEND       ;執行結束，跳到MEND
;--CASE "D"-----;
;--CASE "E"-----;
;--CASE ELSE-----;如果不是上面之數值，則跳到MEND選項，並將P0.1滅，表示接收完畢
MEND:   ;比對結束
        CPL    P0.2        ;使P0.2~P0.4亮，表示資料錯誤
        MOV    R5,#50     ;設定R5=50，延遲50*20ns = 1000ns = 1秒
        CALL   DELAYms    ;呼叫DELAYms副程式
        CPL    P0.1        ;P0.1暗，表示程式比對完畢
        JMP    WAIT       ;跳回WAIT程式繼續等待
;--CASE END-----;

```

圖 3(b). 資料判斷程式

```

;-----以下為副程式-----
;-----
INIT_RS232:                ;RS232轉換副程式
    MOV    TMOD,#20H        ;設定計時器1為模式1，自動載入功能計時器
    MOV    TH1,#0E8H        ;定速率=1200 BPS
    SETB   TR1              ;啓動計時器1
    MOV    SCON,#01010000B  ;設定傳輸協定為模式1，8個資料位元，速率由Timer1決定，
    RET                     ;跳回原程式位置
;-----
DELAYms:                   ;延遲副程式DELAYms
    MOV    R6,#100          ;藉由計算R6、R7的回圈，產生延遲時間20ms
DELAYm :MOV    R7,#100
    DJNZ   R7,$             ;
    DJNZ   R6,DELAYm        ;
    DJNZ   R5,DELAYms       ;總延遲時間由R5決定，時間為(R5*20ms)，5*20ms=100ms=0.1秒
    RET                     ;跳回原程式位置
;-----
    END                     ;程式結束
;-----

```

圖 3(c). 副程式

此程式是以組合語言來設計，可以使用記事本直接進行撰寫，經過組譯與連結，產生可執行檔，便可將此可執行檔燒錄至單晶片裡，進行所需要的工作。

### 3. VB 與 RS232 程式設計

在 Visual Basic 專業版裡，有一個專為設計串列通訊的物件，也就是 MSComm 通訊控制項，只要將它外掛進程式裡，設定好通訊協定，便能使用 RS232 與外界相容之電子設備進行溝通，以下介紹如何設計一個簡易的 VB 與 RS232 的應用程式。

MsComm 通訊控制項屬於特殊功能元件，並不會主動出現在預設工具箱裡，當程式需要此元件時，必須到「專案－設定使用元件－Microsoft Comm Control 6.0」，設定步驟如圖 4 所示。




圖 4. MsComm 通訊控制項設定步驟

引用控制項完成後，便是設定其通訊協定的屬性，如圖 5 所示，設定 RS232 之通訊埠，本程式以設定 2 代表 COM2 為傳輸埠，此數值最大值號碼為 16，除了在這屬性窗設定外，也可以設計在程式裡，如式(1)所示。

`MSComm1.CommPort = 2` (1)




圖 5. 設定通訊埠代號

如圖 6 所示，設定鮑率參數，VB 程式與 89C51 單晶片的組合語言之鮑率參數必須相同，本文是設定 1200bps，其中“1200,N,8,1”，代表“所使用的通訊埠是以每秒 1200Bit 的速度進行傳輸，不作同位位元的檢查，每個位元是 8 個 Bit，停止位元是 1 個 Bit”，同樣地，此參數也可直接設計在程式裡，如式(2)所示。

$$\text{MSComm1.settings} = \text{"1200,N,8,1"} \tag{2}$$




圖 6. 設定鮑率參數

設定好以上兩種基本的參數後，便可開始設計 RS232 溝程序。如圖 7 所示，放置一個文字輸入框，提供使用者輸入指令，按下傳送鈕，便會將文字框裡的指令送至 89C51 單晶片裡，並放置三個控制按鈕，裡頭有專屬的控制指令。如圖 8 所示，為本程式設計說明。




圖 7. 溝通程式範例

```

Project1 - Form1 (程式碼)
Form
Load
Private Sub Form_Load()
    MSCComm1.CommPort = 2           ' 設定RS232通訊埠為COM2
    MSCComm1.Settings = "1200,N,8,1" ' 設定速率參數
    MSCComm1.PortOpen = True       ' 程式一開始便打開通訊埠
End Sub
Private Sub Command1_Click()       ' 傳送按鈕
    Buf = Trim(Text1.Text)         ' 先將Text1裡的資料去掉左右兩邊的空白字元
    MSCComm1.Output = Chr(Buf)     ' 並存進Buf變數裡
    MSCComm1.Output = Chr(Buf)     ' 將Buf變數裡的資料轉成ASCII碼之字元
    MSCComm1.Output = Chr(Buf)     ' 並傳送出去
End Sub
Private Sub Command2_Click()       ' 傳送"1"之按鈕
    MSCComm1.Output = Chr(1)       ' 將"1"的ASCII碼之字元傳送出去
End Sub
Private Sub Command3_Click()       ' 傳送"2"之按鈕
    MSCComm1.Output = Chr(2)       ' 將"2"的ASCII碼之字元傳送出去
End Sub
Private Sub Command4_Click()       ' 傳送"3"之按鈕
    MSCComm1.Output = Chr(3)       ' 將"3"的ASCII碼之字元傳送出去
End Sub
Private Sub Command5_Click()       ' 結束鈕
    MSCComm1.PortOpen = False      ' 將通訊埠關掉
End Sub

```

圖 8. 溝程式說明

當程式與硬體電路準備後之後，便可以開始進行實驗，檢查有無錯誤發生，程式是否如預期運作，本實習電路如圖 9 所示。本程式為基本的單晶片與 PC 的溝程式，為單方向的傳輸，也可以設計成全雙工的形式，使用者將資料傳送至單晶片上，經過處理傳回 PC，顯示給使用者了解，也可設計遠端遙控程式，使單晶片配合現今流行的網際網路，其應用的範圍可說無遠弗屆。




圖 9. VB 與 RS232 實習電路

#### 參考資料

鍾自立、張正賢著，8051 實作與燒錄器製作，宏友書局，2000。

陳龍三著，8051 入門與介面控制，松崗書局，1999。

范逸之、陳立元、賴俊朋著，*Visual Basic 與 RS232 串列通訊控制*，文魁書局，1999。

李齊雄、游國幹著，8051 單晶片微電腦原理與實作，儒林書局，1995。